プログラムのページ

担 当 鈴 木 誠 道

7305 非線形常微分方程式の境界値問題の計算

平野泰彦（電電公社 武蔵野通研）

1. まえがき

非線形常微分方程式の境界値問題の数値解法は反復法によらねばならない．適当な初期条件を与えて，通常の数値解法，たとえば，Runge-Kutta-Gill 法により微分方程式の解を計算し，境界条件を満足するように初期条件を修正してゆくわけである．その修正方法として，線形挿入による代数方程式の区間縮小法が応用されているが，この方法では収束にかなりの反復数を要する．これを改善するには，代数方程式を３点を通る２次曲線で近似して解を得る Muller 法の適用が考えられる[1]．したがって，推定初期値を３個必要とするが，その値が多少はずれても，反復数は著るしく減少する．

2. 計算例

例題として懸垂線の問題をとりあげる．２点間にロープを張ると，ロープは重力によって垂れさがる．そのとき，両端の位置は定まっているが，初期条件である最初の傾斜がわからないので，境界値問題として扱わねばならない．このロープの形状はつぎの非線形微分方程式で表わされる．

$$y''=a\sqrt{1+y'^2}.$$

いま，境界条件を

$$y(0)=0, \qquad y(1)=1$$

とする．ここに，a は定数であって，$a=4$ とする．

このプログラムを図1に示す．SUBROUTINE MULLER より $y'(0)$ の値を受けて，FUNCTION 文の RKGILL で微分方程式を計算し，$y(1)-1$ の値を帰して，$y'(0)$ のよりよい近似値を得る．これを反復して，収束条件を満足すれば演算は打切られる．図２の Runge-Kutta-Gill 法のプログラムは丸め誤差の正しく補償されたプログラムである[2]．また，図３の

```
C     KENSUISEN
      DIMENSION RT(1)
      COMMON Y(2),Q(2),FV(2),AX(20),AY(2,20)
      EXTERNAL FUN
      M=1
      RT(1)=0.01
      E=1.E-7
      CALL MULLER(M,RT,FUN,E)
      WRITE(3,20) (AX(J),(AY(I,J),I=1,2),J=1,20)
   20 FORMAT(3G20.10)
      STOP
      END

      FUNCTION FUN(YY)
      COMMON Y(2),Q(2),FV(2),AX(20),AY(2,20)
      EXTERNAL FUNCVL
      Y(2)=YY
      N=2
      X=0.
      Y(1)=0.
      H=0.05
      M=20
      CALL RKGIIL(N,X,Y,H,M,FUNCVL,Q,FV,AX,AY)
      WRITE(3,10) YY,Y(1)
   10 FORMAT(1H0,10X,6HDY(0)=,F10.7,5X,5HY(1)=,F10.7)
      FUN=Y(1)-1.
      RETURN
      END

      SUBROUTINE FUNCVL(X,Y,F)
      DIMENSION Y(2),F(2)
      F(1)=Y(2)
      F(2)=4.*SQRT(1.+Y(2)*Y(2))
      RETURN
      END
```

図 1 境界値問題として懸垂線の形状を求めるプログラム

```
      SUBROUTINE RKGIIL(N,X,Y,H,M,FUNCVL,Q,FV,AX,AY)
      DIMENSION Y(1),Q(1),FV(1),AX(1),AY(N,1)
      QQ=0.
      DO 10 I=1,N
      Q(I)=0.0
   10 CONTINUE
      CTP=1.7071067812
      CTM=0.29289321881
      HH=H*0.5
      DO 20 K=1,M
      CALL FUNCVL (X,Y,FV)
      DO 2 I=1,N
      FK=H*FV(I)
      R=0.5*FK-Q(I)
      W=Y(I)
      Y(I)=Y(I)+R
      R=Y(I)-W
      Q(I)=Q(I)+3.*R-0.5*FK
    2 CONTINUE
      R=HH-QQ
      W=X
      X=X+R
      R=X-W
      QQ=QQ+R-HH
      CALL FUNCVL(X,Y,FV)
      DO 3 I=1,N
      FK=H*FV(I)
      R=CTM*(FK-Q(I))
      W=Y(I)
      Y(I)=Y(I)+R
      R=Y(I)-W
      Q(I)=Q(I)+3.*R-CTM*FK
    3 CONTINUE
      CALL FUNCVL(X,Y,FV)
      DO 4 I=1,N
      FK=H*FV(I)
      R=CTP*(FK-Q(I))
      W=Y(I)
      Y(I)=Y(I)+R
      R=Y(I)-W
      Q(I)=Q(I)+3.*R-CTP*FK
    4 CONTINUE
      R=HH-QQ
      W=X
      X=X+R
      R=X-W
      QQ=QQ+R-HH
      CALL FUNCVL(X,Y,FV)
      DO 5 I=1,N
      FK=H*FV(I)
      R=(FK-2.*Q(I))/6.
      W=Y(I)
      Y(I)=Y(I)+R
      R=Y(I)-W
      Q(I)=Q(I)+3.*R-0.5*FK
    5 CONTINUE
      AX(K)=X
      DO 6 I=1,N
      AY(I,K)=Y(I)
    6 CONTINUE
   20 CONTINUE
      RETURN
      END
```

図 2 Runge-Kutta-Gill 法のプログラム

Muller 法のプログラムは文献[3]のフローチャートにもとずいてプログラミングしたものである．初期値を指定すると，0.9 RT，1.1 RT および RT が3個の初期値となり，指定がないときは -1，$+1$，0 である．

NEAC 2200/500 計算機を用いて計算した結果を図4に示す．初期値 $y'(0)=0.01$，収束判定値 $\mathrm{EPS}=10^{-7}$ を用いて計算したところ，反復回数は初期値の3回を含めて9回であった．なお，Runge-Kutta-Gill 法を用いたので，x の値は丸め誤差が補償されて小数点2位以下すべて0である．

この方法はつぎの問題にも利用することができる．距離 $l=2$ なる水平な2点間にロープを張り，中央の垂れさがる位置が $h=0.5$ となるように定数 a の値を定める．初期値を中央にとり，$y(0)=0$，$y'(0)=0$ として，$y(1)=0.5$ を満足する a の値を求めればよい．Muller 法を適用したこのプログラムを図5に示す．初期値を $a=1$ にとり，反復回数は最初の3回を含めて5回で収束した．この計算結果を図6に示しておく．

```
      SUBROUTINE MULLER(N,RT,F,E)
      K=0
      DIMENSION RT(1)
   16 K=K+1
      IF(RT(K).EQ.0.) GO TO 10
      R=0.9*RT(K)
      M=1
      GO TO 11
   10 R=-1.
      H=-1.
      M=3
      GO TO 11
   11 FPRT=F(R)
      J=1
      IF(K.EQ.1) GO TO 12
   13 J=J+1
      TEMP=R-RT(J-1)
      FPRT=FPRT/TEMP
   12 IF(K.NE.J) GO TO 13
      GO TO (1,2,3,4,5,6),M
    1 F1=FPRT
      R=1.1*RT(K)
      M=2
      GO TO 11
    2 F2=FPRT
      H=RT(K)-R
      R=RT(K)
      M=5
      GO TO 11
    3 F1=FPRT
      R=1.
      M=4
      GO TO 11
    4 F2=FPRT
      R=0.
      M=5
      GO TO 11
    6 F1=F2
      F2=F3
      F3=FPRT
      GO TO 17
    5 F3=FPRT
      ALAM=-0.5
   17 DELT=1+ALAM
      G=F1*ALAM*ALAM-F2*DELT*DELT+F3*(ALAM+DELT)
      D=G*G-4.*F3*ALAM*DELT*(F1*ALAM-F2*DELT+F3)
      D=G+SIGN(SQRT(ABS(D)),G)
      ALAM=-2.*F3*DELT/D
      H=ALAM*H
      X=R+H
      IF(ABS((X-R)/X)-E) 14,15,15
   15 R=X
      M=6
      GO TO 11
   14 RT(K)=X
      IF(N.NE.K) GO TO 16
      RETURN
      END
```

図 3　Muller 法のプログラム

```
DY(0)=  .0090000      Y(1)= 6.6385245

DY(0)=  .0110000      Y(1)= 6.6523005

DY(0)=  .0100000      Y(1)= 6.6454092

DY(0)= -.9258076      Y(1)= 2.6465947

DY(0)=-1.5123130      Y(1)= 1.6066652

DY(0)=-1.9720629      Y(1)= 1.0874852

DY(0)=-2.0859667      Y(1)=  .9830766

DY(0)=-2.0669438      Y(1)= 1.0001360

DY(0)=-2.0670946      Y(1)= 1.0000000
.5000000000E-01      -.9252448076E-01      -1.646253848
.1000000000E-00      -.1657233547          -1.291481462
.1500000000          -.2225342443          -.9885396539
.2000000000          -.2652370804          -.7252711041
.2500000000          -.2955456025          -.4911103619
.3000000000          -.3146761384          -.2766600101
.3500000000          -.3233964240          -.7331337711E-01
.4000000000          -.3220564194           .1270908233
.4500000000          -.3106023544           .3325958337
.5000000000          -.2885745647           .5514494436
.5500000000          -.2550890380           .7924349166
.6000000000          -.2088019306          1.065223543
.6500000000          -.1478556335          1.380762877
.7000000000          -.6980422209E-01      1.751716171
.7500000000           .2848470451E-01      2.192970619
.8000000000           .1509557334          2.722234821
.8500000000           .3025239438          3.360749469
.9000000000           .4892721612          4.134139785
.9500000000           .7186950770          5.073443913
1.000000000          1.000000029           6.216358561
```

図 4　図1のプログラムの計算結果

```
C     KENSUISEN
      DIMENSION RT(1)
      COMMON A,Y(2),Q(2),FV(2),AX(40),AY(2,40)
      EXTERNAL FUN
      EXTERNAL FUNCVL
      M=1
      RT(1)=1.0
      E=1.E-7
      CALL MULLER(M,RT,FUN,E)
      N=2
      X=0.
      Y(1)=0.
      Y(2)=-AY(2,20)
      H=0.05
      M=40
      WRITE(3,10) RT
   10 FORMAT(4H1 A=,G16.10)
      WRITE(3,15) X,Y
   15 FORMAT(10X,1HX,19X,2HY1,18X,2HY2/3G20.10)
      CALL RKGIIL(N,X,Y,H,M,FUNCVL,Q,FV,AX,AY)
      WRITE(3,20) (AX(J),(AY(I,J),I=1,2),J=1,40)
   20 FORMAT(3G20.10)
      STOP
      END

      FUNCTION FUN(AA)
      COMMON A,Y(2),Q(2),FV(2),AX(40),AY(2,40)
      EXTERNAL FUNCVL
      A=AA
      N=2
      X=0.
      Y(1)=0.
      Y(2)=0.
      H=0.05
      M=20
      CALL RKGIIL(N,X,Y,H,M,FUNCVL,Q,FV,AX,AY)
      WRITE(3,10) A,Y(1)
   10 FORMAT(3H0A=,G16.10,5X,5HY(1)=,G16.10)
      FUN=Y(1)-0.5
      RETURN
      END

      SUBROUTINE FUNCVL(X,Y,F)
      COMMON A
      DIMENSION Y(2),F(2)
      F(1)=Y(2)
      F(2)=A*SQRT(1.+Y(2)*Y(2))
      RETURN
      END
```

図 5　懸垂線の定数を求めるプログラム

```
A= .9000000000        Y(1)= .4812070890

A= 1.100000000        Y(1)= .6077441196

A= 1.000000000        Y(1)= .5430006233

A= .9308540981        Y(1)= .5000201840

A= .9308211703        Y(1)= .4999999784
```

図 6　図5のプログラムの計算結果

3. あとがき

微分方程式の境界値問題における Muller 法の応用について述べた．2つの変数間に関数関係があるときその逆関数の値を求めたいときに，この方法は広く利用できる．たとえば，電子回路の所望の利得に対するある抵抗値を求めたいときなどに利用できる．また，Bessel 関数 $J_0(x)$ が 0.2 なるときの x の値を2個求めたいときは，$M=2$ として1つのプログラムで計算することができる．

参考文献

1) 平野泰彦：コンピュータによる数値計算，日刊工業新聞社，1972，pp. 170.
2) 平野泰彦：常微分方程式の計算における丸め誤差の改善，情報処理，Feb. 1973.
3) S. D. Conte : Elementary Numerical Analysis, McGRAW-HILL Co., 1965, pp. 68.

(昭和48年3月9日　受付)